A Qualitative Logic of Decision

Paul O'Rorke
ororke@ics.uci.edu

Yousri El Fattah
fattah@ics.uci.edu

Technical Report 91–08

January 31, 1991

A Qualitative Logic of Decision[1]

Paul O'Rorke (ororke@ics.uci.edu)
Yousri El Fattah (fattah@ics.uci.edu)

Department of Information and Computer Science
University of California, Irvine, CA 92717
United States of America
Phone: (714) 856-5563
Fax: (714) 856-4056

EXPANDED VERSION

January 31, 1991

Abstract

An important aspect of intelligent behavior is the ability to reason, make decisions, and act in spite of uncertainty. This paper presents a qualitative logic of decision that supports decision-making under uncertainty. To be specific, the paper presents a knowledge representation language based upon subjective Bayesian decision theory that aims to capture some aspects of common-sense reasoning associated with making decisions about actions. The language addresses the problem of describing justifications of rational choices in situations where the alternatives involve trading off potential losses and gains. The logic and an associated qualitative arithmetic are implemented in an efficient PROLOG program. Examples illustrate their use in several concrete decision-making situations.

[1]This paper is a result of research supported in part by the National Science Foundation Grant Number IRI-8813048, McDonnell Douglas Corporation, Douglas Aircraft Company, and the University of California Microelectronics Innovation and Computer Research Opportunities Program.

Contents

List of Figures

List of Tables

1 Introduction

1.1 Motivation

A hallmark of intelligent behavior is the ability to reason, make decisions, and act in spite of uncertainty. Whether our goal is to construct computational models of intelligence or to construct intelligent artifacts, we must have some account of decision-making under uncertainty.

Early AI researchers shunned probability — partly because numerical probabilities are not ordinarily available. In an important early article on knowledge representation, McCarthy and Hayes claimed that "a formalism that required numerical probabilities would be epistemologically inadequate" (McCarthy &Hayes, 1969). Recently, however, there has been increasing interest in subjective Bayesian descriptions of reasoning under uncertainty in AI (see, e.g., Cheeseman, 1985; Pearl, 1988). One reason for this is that several important domains (such as medical diagnosis) have been identified where one can reasonably expect to have the relevant numbers. Another reason is that techniques for qualitative reasoning under uncertainty are beginning to appear.

This paper presents a qualitative logic of decision based upon subjective Bayesian decision theory that aims to capture some aspects of commonsense reasoning associated with making decisions. The logic has the same relation to Bayesian decision theory as qualitative physics has to physics. In cases where abstract reasoning is sufficient and numbers for probabilities and utilitities are not known with certainty, the logic can be used for calculations.

1.2 Deciding Whether to Do an Action

In this section we review the decision-analytic approach to deciding whether to do an operation that may benefit us under certain conditions but may entail a risk of negative consequences if these conditions do not hold (Chung, 1974; North, 1968; Raiffa, 1968). Let P be a predicate, the truth of which we are uncertain about. Assume that the a priori probability of P being true, $Prob(P)$ is p. The odds of P being true is by definition

$$O(P) = \frac{Prob(P)}{Prob(\overline{P})} = \frac{p}{1-p}. \quad (1)$$

Let us assume that we can evaluate the outcomes whether or not we execute the operation, whether or not P is true. In other words, assume that we have a utility function that assigns values to the outcome of executing op when P is true, the outcome of executing op when P is false, the outcome of not executing op when P is true, and the outcome of not executing op when P is false.

$$\begin{aligned}
&utility\ of\ op\ when\ P = U(op, P)\\
&utility\ of\ op\ when\ \overline{P} = U(op, \overline{P})\\
&utility\ of\ \overline{op}\ when\ P\ = U(\overline{op}, P)\\
&utility\ of\ \overline{op}\ when\ \overline{P} = U(\overline{op}, \overline{P})
\end{aligned}$$

The expected utility of action op is the weighted average:

$$EU(op) = U(op, P) \times p + U(op, \overline{P}) \times (1 - p) \tag{2}$$

The expected utility of $\overline{op}$ is given by

$$EU(\overline{op}) = U(\overline{op}, P) \times p + U(\overline{op}, \overline{P}) \times (1 - p) \tag{3}$$

It is reasonable to do op iff its expected utility is as large as that of $\overline{op}$.

$$may(op) \leftrightarrow EU(op) \geq EU(\overline{op}) \tag{4}$$

Another way to look at this is in terms of the odds of P and the change in utility when op is executed. Let us introduce the following notation for the differences between the utility of doing op and the utility of not doing op under the different conditions.

$$\begin{aligned}
\delta_1 &= \Delta U_P(op) = U(op, P) - U(\overline{op}, P)\\
\delta_2 &= \Delta U_{\overline{P}}(op) = U(op, \overline{P}) - U(\overline{op}, \overline{P})
\end{aligned}$$

A positive difference in utility corresponds to a gain, while a negative difference corresponds to a loss incurred by doing op. The inequality in (4) says that op is reasonable iff

$$EU(op) - EU(\overline{op}) \geq 0 \tag{5}$$

This can be expressed in terms of losses and gains:

$$\Delta U_P(op) \times p + \Delta U_{\overline{P}}(op) \times (1 - p) \geq 0 \tag{6}$$



Figure 1: Decision Surface

When p is 1, this means that op is reasonable if it does not lead to a loss when P is true. Otherwise, we can divide by $1-p$ to get the following.

$$O(P) \times \Delta U_P(op) \geq -\Delta U_{\overline{P}}(op) \tag{7}$$

The three dimensional graph in figure 1 shows the decision surface separating situations where one should do the operation from situations where one should not do the operation. We are only concerned with half of three dimensional space since $O(P) \geq 0$. The decision surface corresponds to situations where $\Delta U_{\overline{P}}(op) = -O(P) \times \Delta U_P(op)$. It is reasonable to do op or not at any point on the surface. One should do op in the region marked "+". One should not do op in the region marked "$-$".

Considering any fixed value of the odds of P amounts to taking the intersection of a plane with the decision surface. The result is a line separating the regions with slope equal to $-O(P)$ (see figure 2). If the odds of P is



Figure 2: Decision Curve for Fixed Odds

zero, the line coincides with the horizontal axis. Then, it is certain that P is false, so *op* should be done iff the utility of doing *op* when $\overline{P}$ is greater than the utility of not doing *op* when $\overline{P}$. The vertical axis corresponds to the case when the probability of P is 1. When we are certain that P is true, *op* should be done iff the utility of doing *op* assuming P is greater than the utility of not doing *op* when P.

Note that one should do *op* in the first quadrant since doing *op* yields relatively higher utilities regardless of whether P is true. One should not do *op* in the third quadrant since there are losses associated with doing *op* whether P is true or false. In quadrants two and four, the best decision depends on the odds and utilities. In quadrant four, P supports the decision to do *op*, since there is a gain associated with doing *op* when P is true and a loss associated with doing *op* when P is false. In quadrant two, P weighs against doing *op* since there is a gain associated with doing *op* when P is false and a loss is associated with doing *op* if P is true.

Regardless of the quadrant, the following decision rule holds.

$$should(op) \leftarrow \Delta U_{\overline{P}}(op) > -O(P) \times \Delta U_P(op) \quad (8)$$



Figure 3: Hierarchy of Non-Negative Odds and Utilities

2 An Arithmetic on Qualitative Probabilities and Utilities

In this section we introduce a simple language for expressing qualitative probabilities, odds, costs, and utilities. We also introduce orderings and arithmetic operations that can be used in reasoning about qualitative odds and utilities.

2.1 Qualitative Probabilities and Utilities

There are special constant terms 0, 1, and ∞. Zero and one have their usual interpretation. Infinity is treated in a manner consistent with the role of infinity in the extended reals (see the appendices for details). As shown in figure 3, there are special sorts of variables: σ, β, and π and N. σ stands for a small number strictly greater than zero and less than one while β stands for any number strictly greater than one. π may be used to designate any finite number greater than zero and N designates any nonnegative number (including zero). So N is a completely ambiguous description that may disambiguate to any element of the set $\{0, \sigma, 1, \beta, \infty\}$ given additional information. We use capital nu as a mnemonic: nu for "non-negative." The use of the capital reminds us that (unlike the other sorts) N includes ∞. The type π includes 1 and the subtypes σ and β. Zero, one, and infinity are elements of type N and σ, β, and π are subtypes of type N.

Table 1: Qualitative Addition

$X+Y$	0	σ	1	β	∞
∞	∞	∞	∞	∞	∞
β	β	β	β	β	∞
1	1	β	β	β	∞
σ	σ	π	β	β	∞
0	0	σ	1	β	∞

2.2 Qualitative Comparisons

Several predicates on qualitative probabilities and utilities are useful. Equality is viewed as a co-designation constraint on two descriptions. So an equality between a variable of type π and a variable of type β is satisfied by binding the variable of type π to the variable of type β.

Qualitative orderings are captured by qualitative "less" predicates. Zero is strictly less than all quantities of type π. Quantities of type σ are strictly less than one. One is strictly less than any quantity of type β. Zero, one, and all quantities of type σ, β, and π are strictly less than ∞.

2.3 Arithmetic Operations

Constraints on sums of quantities are captured by a qualitative addition relation summarized in table 1. Note that $\sigma + \sigma$ is ambiguous: it may be any finite quantity greater than zero.

Constraints on products of quantities are captured by a qualitative multiplication relation summarized in table 2. Note that $\sigma \times \beta$ is ambiguous: it may be any quantity greater than zero. Note that we do not follow the standard convention that arbitrarily defines infinity times zero as zero. Instead we make infinity times zero completely ambiguous (N).

Qualitative multiplicative inverses are listed in table 3. Note that $1/0$ is defined to be ∞ and $1/\infty$ is 0. The inverse function maps 1, π, and N into themselves. It maps σ into β and vice versa.

Qualitative division may be defined in terms of multiplication and multiplicative inverse. The resulting qualitative division operation is summarized

Table 2: Qualitative Multiplication

$X \times Y$	0	σ	1	β	∞
∞	N	∞	∞	∞	∞
β	0	π	β	β	∞
1	0	σ	1	β	∞
σ	0	σ	σ	π	∞
0	0	0	0	0	N

Table 3: Qualitative Multiplicative Inverse

X	0	σ	1	β	∞
X^{-1}	∞	β	1	σ	0

in table 4. Note that dividing quantities of type π by zero yields infinity. Zero divided by any positive quantity (including infinity) is (defined to be) zero. Dividing any nonnegative quantity but infinity by infinity also yields zero. Dividing σ or β by itself yields an ambiguous quantity of type π. Zero divided by zero and infinity divided by itself are defined to be ambiguous (N).

Table 4: Qualitative Division

X/Y	0	σ	1	β	∞
∞	∞	∞	∞	∞	N
β	∞	β	β	π	0
1	∞	β	1	σ	0
σ	∞	π	σ	σ	0
0	N	0	0	0	0

3 Qualitative Decisions About Actions

In this section, we describe a qualitative method for deliberation about actions based upon the qualitative calculus of probabilities and utilities described earlier. The method is compatible with a subjective Bayesian view of deliberation. According to Jeffrey (1983), "the Bayesian principle is... *to choose an act of maximum estimated desirability.*"

3.1 Justifiable vs. Necessary Rational Acts

We use the decision theoretic notion of expected utility (Chung, 1974; North, 1968; Raiffa, 1968) to estimate desirability. We use the distinction between indifference and preference to distinguish between justifiable and necessary actions. We shall say that "the action A is justified," "it is reasonable to do action A" or "the qualitative calculus sanctions action A" given constraints on the relevant odds and utilities C if and only if the union of C and the additional constraint

$$\forall x \in \mathit{Alternatives}(A)\,\{EU(A) \geq EU(x)\}$$

is satisfiable.

We shall say that an action is "the only reasonable thing to do," "necessary," or "required by the qualitative calculus" given constraints C on the relevant odds and utilities if and only if

$$\forall x \in \mathit{Alternatives}(A)\,\{EU(A) > EU(x)\}$$

is true in every interpretation of C.

3.2 Decisions Given the Odds of Success

It is interesting to look at the special cases of the *WinLose* tradeoff (Quadrant II of figure 2) that result when we fix the odds of success. If the odds of P is zero, there is no loss to gain ratio that requires the operation, but it is not ruled out if the potential loss is zero or if the potential gain given P is infinite. (See table 5.)

If the odds of p is between zero and one, one should do *op* if there is a gain doing *op* when P is true and the utility does not change when P is false.

Table 5: $O(P) = 0$

$op?$	0	σ	1	β	∞
∞	−	−	−	−	?
β	−	−	−	−	?
1	−	−	−	−	?
σ	−	−	−	−	?
0	?	?	?	?	?

Table 6: $O(P) = \sigma$

$op?$	0	σ	1	β	∞
∞	−	−	−	−	?
β	−	−	−	?	+
1	−	−	−	?	+
σ	−	?	?	?	+
0	?	+	+	+	+

One also should do *op* if the gain is infinite but the potential loss is finite. (See table 6.)

If the odds of p is exactly one, one should do *op* if the potential gain doing op when P is true is strictly greater than the drop in utility from doing *op* when P is false. It is not clear what to do in the cases where the potential gain is the same as the potential loss. (See table 7.)

If P is more likely than not, table 8 applies.

If P is certain then table 9 applies.

3.3 Qualitative Rules for Action

This section presents compact qualitative rules for deciding whether to perform a given act. $Should(op)$ is intended to be true in situations corresponding to points in the decision region marked + in figures 1 and 2. $Should(\overline{op})$ is intended to be true in the decision region marked −.

Table 7: $O(P) = 1$

op?	0	σ	1	β	∞
∞	−	−	−	−	?
β	−	−	−	?	+
1	−	−	?	+	+
σ	−	?	+	+	+
0	?	+	+	+	+

Table 8: $O(P) = \beta$

op?	0	σ	1	β	∞
∞	−	−	−	−	?
β	−	?	?	?	+
1	−	?	+	+	+
σ	−	?	+	+	+
0	?	+	+	+	+

Table 9: $O(P) = \infty$

op?	0	σ	1	β	∞
∞	?	?	?	?	?
β	?	+	+	+	+
1	?	+	+	+	+
σ	?	+	+	+	+
0	?	+	+	+	+

$$
\begin{array}{l}
WinWin: \; should(op) \leftarrow \delta_1 > 0, \; \delta_2 > 0. \\
WinLose: \; gain = \delta_1, \; loss = -\,\delta_2, \; odds = O(P) \leftarrow \delta_1 \geq 0 \geq \delta_2. \\
LoseWin: \; gain = \delta_2, \; loss = -\,\delta_1, \; odds = O(\overline{P}) \leftarrow \delta_2 \geq 0 \geq \delta_1. \\
Y_1: \; should(op) \leftarrow gain = \infty, \; loss \neq \infty, \; odds > 0. \\
Y_2: \; should(op) \leftarrow gain > 0, \; loss \neq \infty, \; odds = \infty. \\
Y_3: \; should(op) \leftarrow gain > 0, \; loss = 0, \; odds > 0. \\
Y_4: \; should(op) \leftarrow gain \geq loss, \; odds > 1. \\
Y_5: \; should(op) \leftarrow gain > loss, \; odds \geq 1. \\
N_1: \; should(\overline{op}) \leftarrow gain \neq \infty, \; loss = \infty, \; odds \neq \infty. \\
N_2: \; should(\overline{op}) \leftarrow gain \neq \infty, \; loss > 0, \; odds = 0. \\
N_3: \; should(\overline{op}) \leftarrow gain = 0, \; loss > 0, \; odds \neq \infty. \\
N_4: \; should(\overline{op}) \leftarrow loss \geq gain, \; odds > 1. \\
N_5: \; should(\overline{op}) \leftarrow loss > gain, \; odds \geq 1. \\
LoseLose: \; should(\overline{op}) \leftarrow 0 > \delta_1, \; 0 > \delta_2.
\end{array}
$$

Figure 4: Qualitative Rules for Action Based on Linear Inequalities

Special cases in which it is expected that one will do op are given in figure 4. This rule set is incomplete but nevertheless the rules specify a valuable set of conditions under which an operation is reasonable or expected. The $WinWin$ rule covers situations where doing *op* is advantageous whether or not P is true. In these situations, the odds of P are irrelevant and one should obviously do the operation. The $LoseLose$ rule covers situations where doing *op* leads to losses whether or not P is true. Again, in these cases the odds of P are irrelevant and one should not do *op*.

The $WinLose$ and $LoseWin$ rules exploit symmetries to get a more concise collection of decision rules covering quadrants four and two, respectively. Rule $Y1$ says when to do an action that leads to overwhelming gains. Rule $N1$ says to avoid actions that lead to overwhelming losses. Rules $Y2$ and $N2$ apply to situations where one is certain to gain (or lose). Rules $Y3$ and $N3$ deal with situations where there is no prospect of loss (gain). $Y4$ and $Y5$ say one should do *op* if the potential gain outweighs the potential loss and the odds of the gain are better than even. One should not do *op* when it has a less than even chance of success and the downside outweighs the potential gain ($N4$, $N5$).

4 Examples

In this section, we illustrate the qualitative logic of decision by applying it to several examples involving tradeoffs between potential gains and losses. The first one is based on Jeffrey's "the right wine" example (1983, pp. 3-4). This second one illustrates the use of infinite utilities using a well known religious argument: Pascal's wager.

4.1 Deciding Which Wine to Bring

> *"The dinner guest who is to provide the wine has forgotten whether chicken or beef is to be served. He has no telephone, has a bottle of white and a bottle of red, and can only bring one of them (in an oversized pocket) since he is going by bicycle."*

Assume that the relevant odds and utilities are as shown in figure 5. The absent-minded guest believes the chances of chicken and beef are even. Red



Figure 5: Decision Analysis for the Wine Example

wine with beef and white wine with chicken are rated as equally good outcomes ($\infty \neq G > 0$). He is indifferent to the possibility of drinking red wine with chicken but would prefer to avoid bringing white wine if beef is to be served ($-\infty \neq -B < 0$).

The constraints in this case are: $\delta_1 = G - -B = G + B$, $\delta_2 = 0 - G = -G$, and $Odds = 1$. Regardless of the exact values of G and B, the guest should bring the red wine since this is the better choice. In terms of expected utility, $G > G - B$. Note that this is a special case of the $WinLose$ situation and rule Y_5 in figure 4 applies.

4.2 Pascal's Wager

In his *Pensées* (Pascal, 1966) Pascal argued that rational individuals should choose to live their lives as if God exists, even though it may not be possible to determine whether God actually exists.

> *Let us then examine this point and say, "God is, or He is not." ... What will you wager? ... Let us weigh the gain and loss of wagering that God is ... there is here an infinity of an infinitely happy life to gain, a chance of gain against a finite number of chances*



Figure 6: Decision Analysis for Pascal's Wager

> *of loss, and what you stake is finite ... every player stakes a certainty to gain an uncertainty, and yet he stakes a finite certainty to gain a finite uncertainty, without transgressing against reason ... the uncertainty of the gain is proportioned to the certainty of the stake according to the proportion of the chances of gain and loss. ... And so our proposition is of infinite force, when there is the finite to stake in a game where there are equal risks of gain and loss, and the infinite to gain. This is demonstrable; and if men are capable of truths, this is one. ...*

Following (Chimenti, 1990)[2] Let p be the probability that God exists in Pascal's sense. In other words, in this example, the proposition P is that God exists, God cares about each individual's behavior, and God will reward (punish) behavior he (dis)approves of in an eternal afterlife. Figure 6 illustrates a simplified version of Chimenti's analysis of the decision. The question is whether to live a life pleasing to God (this is the choice labelled "be nice" in the figure) or to live a life pleasing to oneself (and "be naughty")

[2]We simplify Chimenti's analysis a bit by ignoring complications (such as the possibility of "quixotic payoffs") that do not affect the conclusion of the decision analysis.

instead.

If we're nice we shoulder the burden ($-B \leq 0$) of a disciplined life but if God exists we obtain the "joys of heaven" ($J \geq 0$). If we're naughty we obtain the "gratifications of life" ($G \geq 0$) but if God exists, we suffer the "anguish of Hell" ($-A \leq 0$).

The expected utility of a virtuous life is $p \times (J - B) + (1 - p) \times -B = p \times J - B$. The expected utility of sin is $p \times (G - A) + (1 - p) \times G = G - p \times A$.

Looking at this is in terms of odds, losses and gains, the net gain gotten by refraining from sin if God exists is $\delta_1 = J - B - (G - A) = J + A - B - G$. The loss when God does not exist is $\delta_2 = -B - G$.

In terms of our qualitative calculus, this is a special case of rule Y_1 in figure 4. Pascal assumes that $J = \infty = A$ but that $B \neq \infty \neq G$. The relevant inequality is then $\beta \leq odds \times \infty$. This is true (for some value of the product) for all possible (non-negative) *odds* including $odds = 0$. We interpret this to mean that no matter what the odds of God's existence, it is reasonable to act as if He existed.

Fortunately, the ambiguity of $0 \times \infty$ makes it possible to be a hedonistic atheist without being irrational. However, mere agnostics are condemned to being good or being irrational. Any value of *odds* consistent with the constraint $0 < odds$ implies the strict inequality $\beta < odds \times \infty$. So if one is the least bit uncertain about God's (non) existence, one should live as if He does exist.

5 Related Work

Our arithmetic for combining constraints has close relatives in work on qualitative reasoning about the physical world (Weld &de Kleer, 1990). The main difference is that we employ different constraints and landmarks since we deal in odds and utilities instead of purely physical quantities like pressures and levels of liquids. The economic view of rational deliberation about action taken here was advocated in an invited talk by Jon Doyle (Doyle, 1990). It is also inspired by advanced work on decision theory in AI involving medical applications (see, for example, Langlotz, Fagan, Tu &Shortliffe, 1987; Langlotz &Shortliffe, 1989; Langlotz, Shortliffe &Fagan, 1986a; Langlotz, Shortliffe &Fagan, 1986b).

The most closely related domain-independent work is Wellman's research on *qualitative probabilistic networks for planning* (Wellman, 1988). Wellman presents a method for determining admissible plans based on graphical manipulations of influence diagrams. Wellman's method is similar to our work in that both aim to identify conclusions that do not depend on the exact values of quantities. It is important to note, however, that Wellman's scheme was intended to form only part of a comprehensive planning program. The kinds of qualitative influences exploited by Wellman are not sufficient to resolve true tradeoffs. Our work breaks thru the "equivalence up to trade-offs" by taking advantage of additional constraints on odds and utilities.

6 Conclusion

Real world tasks often involve goals of varying importance and they often require trading off costs and benefits. For example, in studying examples of diagnostic planning with our industrial collaborators, we find that technicians often must decide whether to verify that a suspected fault is actually present. Sometimes it is more cost-effective simply to replace the potentially faulty component, as when confirmation is expensive but repair is inexpensive. The logic presented here enables us to automate this sort of diagnostic decision-making even when we don't have exact figures for the relevant odds, costs, and benefits.

If the relevant numbers are available, we take advantage of them. But even then, the abstractions afforded by the qualitative calculus can be useful in stripping away inessential details of particular examples. Like Wellman, we also aim to identify minimal assumptions necessary for results so as to provide more coherent and compelling explanations than those generated under complete information. In fact, our main motivation for developing a qualitative logic of decision is to extend explanation-based learning to domains involving uncertainty and goals of various priorities (see O'Rorke &El Fattah, 1991).

Acknowledgements

MATHEMATICA was useful in exploring the geometry of the decision surface. Thanks to Brian Skyrms and Rina Dechter for advice and for pointers to relevant papers and books. Thanks to Margaret Elliott, Steve Morris, David Aha, Dennis Kibler, Ruediger Wirth, and the other members of the machine learning community at UCI for additional help and feedback. Margaret Elliott implemented a qualitative logic of decision in LPA's MacPROLOG and ran computational experiments that pointed to some early bugs in interpreting ambiguities. Margaret Elliott also helped incorporate negative numbers into the qualitative arithmetic.

A Extended Arithmetic on Odds and Utilities

Here we describe an extension of the qualitative arithmetic on odds and utilities that allows negative utilities. This arithmetic adapts the usual conventions associated with the extended real numbers (Royden, 1968; Rudin, 1976).

The real number system $\Re$ is extended by adding positive and negative infinities. The result, $\Re \cup \{-\infty, +\infty\}$, is not a field but the following conventions are usually adopted.

$$\Re \cup \{-\infty, +\infty\}$$
$$\infty + \infty = \infty, -\infty - \infty = -\infty;$$
$$\infty \times (\pm\infty) = \pm\infty, -\infty \times (\pm\infty) = \mp\infty;$$
$$\forall x \in \Re \ldots$$
$$-\infty < x < \infty;$$
$$x + \infty = \infty, x - \infty = -\infty, \frac{x}{\infty} = \frac{x}{-\infty} = 0;$$
$$x > 0 \Rightarrow x \times \infty = \infty, x \times (-\infty) = -\infty;$$
$$x < 0 \Rightarrow x \times \infty = -\infty, x \times (-\infty) = \infty.$$

The arbitrary convention that $0 \times \infty = 0$ is usually adopted as well. For our purposes, however, it is better to make this product ambiguous. Ambiguities are allowed if we view arithmetic operations as defining relations rather than well-defined functions. While $\infty - \infty$ is usually left undefined, we make the difference ambiguous, allowing it to take on any value from $-\infty$ to $+\infty$. While division by zero is usually left undefined, we adopt the convention that $0/0$ is ambiguous but dividing anything except 0 by 0 yields $\pm\infty$.

Table 10: Qualitative Addition With Negatives

$X+Y$	$-\infty$	$-\beta$	-1	$-\sigma$	0	σ	1	β	∞
∞	$\pm N$	∞	∞	∞	∞	∞	∞	∞	∞
β	$-\infty$	$\{0,\pm\pi\}$	π	π	β	β	β	β	∞
1	$-\infty$	$-\pi$	0	σ	1	β	β	β	∞
σ	$-\infty$	$-\pi$	$-\sigma$	$\{0,\pm\sigma\}$	σ	π	β	β	∞
0	$-\infty$	$-\beta$	-1	$-\sigma$	0	σ	1	β	∞
$-\sigma$	$-\infty$	$-\beta$	$-\beta$	$-\pi$	$-\sigma$	$\{0,\pm\sigma\}$	σ	π	∞
-1	$-\infty$	$-\beta$	$-\beta$	$-\beta$	-1	$-\sigma$	0	π	∞
$-\beta$	$-\infty$	$-\beta$	$-\beta$	$-\beta$	$-\beta$	$-\pi$	$-\pi$	$\{0,\pm\pi\}$	∞
$-\infty$	$-\infty$	$-\infty$	$-\infty$	$-\infty$	$-\infty$	$-\infty$	$-\infty$	$-\infty$	$\pm N$

Table 11: Qualitative Additive Inverse With Negatives

X	$-\infty$	$-\beta$	-1	$-\sigma$	0	σ	1	β	∞
$-X$	∞	β	1	σ	0	$-\sigma$	-1	$-\beta$	$-\infty$

Table 12: Qualitative Subtraction With Negatives

$X+Y$	$-\infty$	$-\beta$	-1	$-\sigma$	0	σ	1	β	∞
∞	∞	∞	∞	∞	∞	∞	∞	∞	$\pm N$
β	∞	β	β	β	β	π	π	$\{0,\pm\pi\}$	$-\infty$
1	∞	β	β	β	1	σ	0	$-\pi$	$-\infty$
σ	∞	β	β	π	σ	$\{0,\pm\sigma\}$	$-\sigma$	$-\pi$	$-\infty$
0	∞	β	1	σ	0	$-\sigma$	-1	$-\beta$	$-\infty$
$-\sigma$	∞	π	σ	$\{0,\pm\sigma\}$	$-\sigma$	$-\pi$	$-\beta$	$-\beta$	$-\infty$
-1	∞	π	0	$-\sigma$	-1	$-\beta$	$-\beta$	$-\beta$	$-\infty$
$-\beta$	∞	$\{0,\pm\pi\}$	$-\pi$	$-\pi$	$-\beta$	$-\beta$	$-\beta$	$-\beta$	$-\infty$
$-\infty$	$\pm N$	$-\infty$	$-\infty$	$-\infty$	$-\infty$	$-\infty$	$-\infty$	$-\infty$	$-\infty$

Table 13: Qualitative Multiplication With Negatives

$X \times Y$	$-\infty$	$-\beta$	-1	$-\sigma$	0	σ	1	β	∞
∞	$-\infty$	$-\infty$	$-\infty$	$-\infty$	$\pm N$	∞	∞	∞	∞
β	$-\infty$	$-\beta$	$-\beta$	$-\pi$	0	π	β	β	∞
1	$-\infty$	$-\beta$	-1	$-\sigma$	0	σ	1	β	∞
σ	$-\infty$	$-\pi$	$-\sigma$	$-\sigma$	0	σ	σ	π	∞
0	$\pm N$	0	0	0	0	0	0	0	$\pm N$
$-\sigma$	∞	π	σ	σ	0	$-\sigma$	$-\sigma$	$-\pi$	$-\infty$
-1	∞	β	1	σ	0	$-\sigma$	-1	$-\beta$	$-\infty$
$-\beta$	∞	β	β	π	0	$-\pi$	$-\beta$	$-\beta$	$-\infty$
$-\infty$	∞	∞	∞	∞	$\pm N$	$-\infty$	$-\infty$	$-\infty$	$-\infty$

Table 14: Qualitative Multiplicative Inverse With Negatives

X	$-\infty$	$-\beta$	-1	$-\sigma$	0	σ	1	β	∞
X^{-1}	0	$-\sigma$	-1	$-\beta$	$\pm\infty$	β	1	σ	0

Table 15: Qualitative Division With Negatives

X/Y	$-\infty$	$-\beta$	-1	$-\sigma$	0	σ	1	β	∞
∞	$\pm N$	$-\infty$	$-\infty$	$-\infty$	$\pm\infty$	∞	∞	∞	$\pm N$
β	0	$-\pi$	$-\beta$	$-\beta$	$\pm\infty$	β	β	π	0
1	0	$-\sigma$	-1	$-\beta$	$\pm\infty$	β	1	σ	0
σ	0	$-\sigma$	$-\sigma$	$-\pi$	$\pm\infty$	π	σ	σ	0
0	0	0	0	0	$\pm N$	0	0	0	0
$-\sigma$	0	σ	σ	π	$\pm\infty$	$-\pi$	$-\sigma$	$-\sigma$	0
-1	0	σ	1	β	$\pm\infty$	$-\beta$	-1	$-\sigma$	0
$-\beta$	0	π	β	β	$\pm\infty$	$-\beta$	$-\beta$	$-\pi$	0
$-\infty$	$\pm N$	∞	∞	∞	$\pm\infty$	$-\infty$	$-\infty$	$-\infty$	$\pm N$

B PROLOG Implementation

```
% Decisions, decisions...
% Version 4. Works with Qualitative Arithmetic version 4.
% Primary improvement over version one is incorporation of numbers in
% addition to 0,sigma,1,beta, and infinity.
% This version can do the example O=3,D=1750,Md=150.
% Handles Pascal's wager:
%   D=infinity, Md=beta yields all possible values for odds except zero.
% (Note generate and test needed to handle all possible values
% of ambiguous quantities like pi*beta.)

% Decisions about whether to do op.
% O is the prior Odds of some Predicate P being true.
% Md is the negative of the expected utility of doing op when not P minus the
%    expected utility of not doing op when not P.
% D is the expected utility of doing op when P minus
%    the expected utility of not doing op when P.
may_do_op(O,D,Md):- qmult(O,D,Prod),qless_or_equal(Md,Prod).
may_not_do_op(O,D,Md):- qmult(O,D,Prod),qless_or_equal(Prod,Md).
ambivalent(O,D,Md):- may_do_op(O,D,Md),may_not_do_op(O,D,Md).
should_do_op(O,D,Md):- non_negative(O),non_negative(D),non_negative(Md),
                                   forall(qmult(O,D,Prod),qless(Md,Prod)).
should_not_do_op(O,D,Md):- qmult(O,D,Prod),qless(Prod, Md).
% should_not_do_op(O,D,Md):- may_not_do_op(O,D,Md),not ambivalent(O,D,Md).
poss_situation(O,D,Md):-non_negative(O), non_negative(Md), non_negative(D).
simple_shoulds(L):-setof([O,D,Md],
                                     (poss_situation(O,D,Md),
                                      should_do_op(O,D,Md)),L).
simple_should_nots(L):-setof([O,D,Md],(poss_situation(O,D,Md),
                              should_not_do_op(O,D,Md)),L).
simple_may_or_may_nots(L):-setof([O,D,Md],(poss_situation(O,D,Md),
                              ambivalent(O,D,Md)),L).
buggy(O,M,D):-should_do_op(O,M,D),should_not_do_op(O,M,D).

member(X,L):-on(X,L).           % for LPA MacProlog
```

```
% LPA PROLOG implementation of a qualitative arithmetic on odds and utilities
%  by Paul O'Rorke & Margaret Elliott.
% Zero and One (and minus One) are the only concrete landmarks
% in the quantity spaces.
% One is significant because it marks even odds.
% -infinity<-beta<-1<-sigma<0<sigma<1<beta<infinity.
% Concrete numbers may be used in addition to 0&1.
% Incorporates negatives.
% Compatible with most of the std conventions on the extended real numbers
%   (see Rudin "Principles of Mathematical Analysis"
%          and Royden "Real Analysis").
% Tested against tables 1/31/91.
% The strategy here is to avoid throwing away information
% contained in given numbers.
% Unless indicated otherwise, the following statements are certainly true.
% In the current approach, ambiguities are represented
% as alternative interpretations.
%   it might be better to introduce ambiguous constants in a hierarchy.
%   nu for arbitrary non-negative numbers.
%   pqnum for undetermined elements of {sigma,1,beta}.
%   pqnum is used in place of pi here to avoid colliding
%   with the built-in pi relation.
% non_negative(nu).
non_negative(0).
% positive(pqnum).
pqnum(sigma).
pqnum(1).
pqnum(beta).
pqnum(X):-number(X),0<X.
nqnum(-X):-pqnum(X).
positive(X):-pqnum(X).
positive(infinity).
non_negative(X):-positive(X).
negative(-X):-positive(X).
negative(-1). % negative(-X) fails on negative(-1) without this. Why?

qnum(0).
qnum(Z):-positive(Z).
qnum(Z):-negative(Z).
```

```
% Qualitative Equivalence and Equality.
qequiv(X,sigma):-number(X),0<X,X<1.
qequiv(X,-sigma):-number(X),Y is -1*X,qequiv(Y,sigma).
qequiv(X,beta):-number(X),1<X.
qequiv(X,-beta):-number(X),Y is -1*X,qequiv(Y,beta).
qequal(X,X).

% Qualitative Ordering.
% Here we only order the bottom elements
% of the hierarchical quantity space.
% Next three lines same as
% qsuccessor(-X,-Y):-positive(X),positive(Y),qsuccessor(Y,X).
qsuccessor(-infinity,-beta).
qsuccessor(-beta,-1).
qsuccessor(-1,-sigma).
qsuccessor(-sigma,0).
qsuccessor(0,sigma).
qsuccessor(sigma,1).
qsuccessor(1,beta).
qsuccessor(beta,infinity).
qsuccessor(X,Y):-number(X),qequiv(X,Z),qsuccessor(Z,Y).
qless(X,Y):- number(X),number(Y),X<Y.
qless(X,Y):- qsuccessor(X,Y).
qless(X,Y):- qsuccessor(X,Z), qless(Z,Y).
qless_or_equal(X,Y):- number(X),number(Y),X=<Y.
qless_or_equal(X,Y):- qequal(X,Y).
qless_or_equal(X,Y):- qequiv(X,Z), qequal(Z,Y).
qless_or_equal(X,Y):- qequiv(Y,Z), qequal(X,Z).
qless_or_equal(X,Y):- qless(X,Y).

% Qualitative Addition.
qadd(X,Y,Z):- number(X), number(Y), Z is X+Y.
qadd(X,Y,Z):- number(X), not number(Y), qequiv(X,XX), qadd(XX,Y,Z).
qadd(X,Y,Z):- not number(X), number(Y), qequiv(Y,YY), qadd(X,YY,Z).
% qadd(-X,-Y,-Z):-positive(X), positive(Y),qadd(X,Y,Z). % Doesn't work.  Why?
qadd(Mx,My,Mz):-negative(Mx),qminus(X,Mx),
                          negative(My),qminus(Y,My),
                          qadd(X,Y,Z),qminus(Z,Mz).
```

```
% qadd(-X,0,-X):-positive(X). % Doesn't work.  Why?
qadd(Mx,0,Mx):-negative(Mx).
%     Addition is commutative, so we need only write laws for the case
%     when the first argument is qualitatively
%     less than or equal to the second argument.
qadd(X,Y,Z):- qless(Y,X), qadd(Y,X,Z).  % Switch X and Y to get the other cases.
qadd(-infinity,infinity,Z):- qnum(Z).  % ambiguous.  undefined in Royden.
qadd(-infinity,X,-infinity):-qnum(X), not X=infinity.
% -sigma if -1-sigma<-beta<-1;
% -1 if -beta=-1-sigma<-1;
% -beta if -beta<-1-sigma<-1.
qadd(-beta,sigma,-X):-pqnum(X).
% -sigma if -2<-beta<-1; -1 if -2=-beta<-1; -beta if -beta<-2<-1.
qadd(-beta,1,-X):-pqnum(X).
% -beta1+beta2<-1 if beta2<beta1-1; =-1 if beta2=beta1-1.
qadd(-beta,beta,-X):-pqnum(X).
qadd(-beta,beta,0).
qadd(-beta,beta,X):-pqnum(X).
qadd(-1,sigma,-sigma).
qadd(-1,1,0).                          % redundant.
qadd(-1,beta,X):-pqnum(X). % sigma if 1<beta<2; 1 if beta=2; beta if 2<beta
qadd(-sigma,sigma,-sigma).
qadd(-sigma,sigma,0).
qadd(-sigma,sigma,sigma).
qadd(-sigma,1,sigma).
qadd(-sigma,beta,X):-pqnum(X).
qadd(0,Y,Y):-qnum(Y).
% qadd(sigma,sigma,pqnum). % ambiguous. following interpretations possible.
qadd(sigma,sigma,X):- pqnum(X).
qadd(sigma,1,beta).
qadd(sigma,beta,beta).
qadd(1,1,beta).
qadd(1,beta,beta).
qadd(beta,beta,beta).
qadd(X,infinity,infinity):-qnum(X).

% Qualitative Additive Inverse.
qminus(infinity,-infinity).
qminus(beta,-beta).
```

```
qminus(1,-1).
qminus(sigma,-sigma).
qminus(0,0).
qminus(-sigma,sigma).
qminus(-1, 1).
qminus(-beta, beta).
qminus(-infinity,infinity).

% Qualitative Subtraction.
qsub(X,Y,Z):-qminus(Y,MY), qadd(X,MY,Z).

% Qualitative Multiplication.
qmult(X,Y,Z):- number(X),number(Y), Z is X*Y.
qmult(X,Y,Z):- number(X), not number(Y),
                        qequiv(X,XX), qmult(XX,Y,Z).
qmult(X,Y,Z):- not number(X), number(Y),
                        qequiv(Y,YY), qmult(X,YY,Z).
% qmult(-X,-Y,Z):- positive(X), positive(Y),qmult(X,Y,Z). % Doesn't work. Why?
qmult(Mx,My,Z):- negative(Mx),qminus(Mx,X),
                          negative(My),qminus(My,Y),qmult(X,Y,Z).
% qmult(-X,Y,-Z):- positive(X),positive(Y),qmult(X,Y,Z). % Doesn't work. Why?
qmult(Mx,Y,Mz):- qminus(Mx,X),positive(X),positive(Y),
                          qmult(X,Y,Z),qminus(Z,Mz).
%     Multiplication is commutative, so we need only write laws
%     for the case when the  first arg is
%     qualitatively less than or equal to the second.
qmult(X,Y,Z):- qless(Y,X), qmult(Y,X,Z). % Switch args to get other cases.
qmult(-infinity,0,Z):- qnum(Z). % zero times infinity indeterminate.
qmult(0,Y,0):- positive(Y), qless(Y,infinity).
qmult(X,0,0):- negative(X).
qmult(0,infinity,Z):- qnum(Z). % zero times infinity indeterminate.
qmult(sigma,sigma,sigma).
qmult(sigma,1,sigma).
% qmult(sigma,beta,pqnum). the following are 3 interpretations.
qmult(sigma,beta,X):-pqnum(X).
qmult(1,Y,Y):-qless_or_equal(1,Y).
qmult(beta,beta,beta).
qmult(X,infinity,infinity):-positive(X).
```

```
% Qualitative Multiplicative Inverse.
% Made the inverse of 0 fail in the past.
% Now treating it as plus or minus infinity.
% PROLOG may collapse -0 and 0
%  but this is what we want at present (we have no epsilons).
qinverse(0,infinity).
qinverse(0,-infinity).
qinverse(X,Y):- number(X), not X=0, not number(Y), Y is 1/X.
qinverse(X,Y):- number(Y), not Y=0,  not number(X), X is 1/Y.
qinverse(sigma,beta).
qinverse(1,1).  %  Redundant but keep for generation of solutions.
qinverse(beta,sigma).
qinverse(infinity,0).
qinverse(-infinity,0).
% qinverse(-X,-Y):- pqnum(X), qinverse(X,Y). % Doesn't work for Y=1.
qinverse(Mx,My):-negative(Mx),qminus(Mx,X),negative(My),
                           qminus(My,Y),qinverse(X,Y).

% Qualitative Division.
qdiv(X,Y,Z):- qinverse(Y,Yinv), qmult(X,Yinv,Z).
```

References

Cheeseman, P. (1985). In defense of probability. *International Joint Conference on Artificial Intelligence* (pp. 1002-1009). Los Angeles, CA: Morgan Kaufmann.

Chimenti, F. A. (December 1990). Pascal's wager: A decision-theoretic approach. *Mathematics Magazine*, pp. 321-325.

Chung, K. L. (1974). *Elementary probability theory with stochastic processes* (second ed.). New York, NY: Springer Verlag.

Doyle, J. (1990). Rationality and its roles in reasoning (extended abstract). *Proceedings of the Eighth National Conference on Artificial Intelligence* (pp. 1093-1100). Boston, MA: AAAI Press; MIT Press.

Jeffrey, R. C. (1983). *The logic of decision* (second ed.). Chicago: The University of Chicago Press.

Langlotz, C. P., Fagan, L. M., Tu, S. W., & Shortliffe, E. H. (1987). A therapy planning architecture that combines decision theory and artificial intelligence. *Computers and Biomedical Research*, *20*, 279-303.

Langlotz, C. P., & Shortliffe, E. H. (1989). Logical and decision-theoretic methods for planning under uncertainty. *AI Magazine*, pp. 39-47.

Langlotz, C. P., Shortliffe, E. H., & Fagan, L. M. (1986a). Computer-based explanation of decision analysis (abstract presented at the Eighth Annual Meeting of the Society for Medical Decision Making, October 19-22, 1986, Chicago, IL). *Medical Decision Making*, *6*(4), 277.

Langlotz, C. P., Shortliffe, E. H., & Fagan, L. M. (1986b). Using decision theory to justify heuristics. *Proceedings of the National Conference on Artificial Intelligence* (pp. 215-219). Philadelphia, PA: Morgan-Kaufmann.

McCarthy, J., & Hayes, P. (1969). Some philosophical problems from the standpoint of artificial intelligence. *Machine Intelligence.* (pp. 463-502). Edinburgh, UK: Edinburgh University Press.

North, D. W. (1968). A tutorial introduction to decision theory. *IEEE Transactions on Systems Science and Cybernetics*, *SSC-4*(3), O'Rorke, P., & El Fattah, Y. (1991). *Learning qualitative decision rules and preferences*, (DRAFT submitted to IJCAI-91). Department of Information and Computer Science, University of California, Irvine.

Pascal, B. (1966). *The Thoughts of Blaise Pascal.* New York: Doubleday Company, Inc.

Pearl, J. (1988). *Probabilistic reasoning in intelligent systems: Networks of plausible inference.* San Mateo, CA: Morgan Kaufmann.

Raiffa, H. (1968). *Decision analysis: Introductory lectures on choices under uncertainty* (first ed.). New York, NY: Random House.

Royden, H. L. (1968). The extended real numbers. *Real Analysis.* (pp. 34). New York: MacMillan Publishing Company, Inc.

Rudin, W. (1976). The extended real number system. *Principles of Mathematical Analysis.* (pp. 11-12). New York: McGraw-Hill Book Company.

Weld, D. S., & de Kleer, J. (Eds.). (1990). Chapter 3: Mathematical aspects of qualitative reasoning. In *Readings in Qualitative Reasoning About Physical Systems.* San Mateo, CA: Morgan Kaufmann.

Wellman, M. P. (1988). Qualitative probabilistic networks for planning under uncertainty. In J. F. Lemmer, & L. N. Kanal (Eds.), *Uncertainty in Artificial Intelligence.* Amsterdam, Holland: Elsevier.